

## PIE CHART

### PURPOSE

Generates a pie chart.

### DESCRIPTION

A pie chart is a graphical data analysis technique for summarizing the distributional information of a variable. It is a circular plot consisting of wedges where the size of each wedge is proportional to the frequency (= number of observations) in that wedge. The plot is to be read clockwise (where the first wedge is at 9 o'clock).

If a single variable is specified, DATAPLOT divides the values into frequency classes in the same manner as for a histogram. The histogram and the pie chart have the same information except the histogram has bars at the data values (where the height of the bar is proportional to the number of observations in the class), whereas the pie chart has wedges (where the area of the wedge is proportional to the number of observations in the class).

If two variables are specified, the first variable contains pre-computed frequencies and the second variable is a group identifier. This second form is more commonly used.

### SYNTAX 1

PIE CHART <x> <SUBSET/EXCEPT/FOR qualification>

where <x> is the variable of raw data values;  
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used when you have raw data only.

### SYNTAX 2

PIE CHART <y> <x> <SUBSET/EXCEPT/FOR qualification>

where <y> is the variable of pre-computed frequencies;  
<x> is the variable of group identifiers;  
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used when you have pre-computed frequencies at each data value.

### EXAMPLES

```
PIE CHART X
PIE CHART TEMP SUBSET TEMP > 0
PIE CHART F X SUBSET X > 2
PIE CHART COUNTS STATE
```

### NOTE 1

Each wedge is drawn with a common set of attributes. The attributes of the wedge borders are set with the LINE, LINE COLOR, and LINE THICKNESS commands (typically they are all set the same). The attributes of the interior are set with the various REGION commands. Any labels for the wedges must be set with the LEGEND or TEXT commands. The CROSS HAIR command can help in positioning labels. The program example below shows how to set the attributes. DATAPLOT does not support features such as 3d pie charts or exploding slices that are common in many business graphics programs.

### NOTE 2

Although pie charts are popular in business graphics, they are generally a poor graphics technique. See the book listed in the REFERENCE section below for more information.

### NOTE 3

For the one variable form of the command, DATAPLOT uses a class width of 0.3 times the standard deviation of the variable. Use the CLASS WIDTH to override this default. DATAPLOT also tends to generate a large number of zero frequency classes at the lower and upper tails. The CLASS LOWER and CLASS UPPER commands can be used to set lower and upper limits for the classes.

### DEFAULT

None

### SYNONYMS

None

**RELATED COMMANDS**

HISTOGRAM	=	Generates a histogram.
FREQUENCY PLOT	=	Generates a frequency plot.
PERCENT POINT PLOT	=	Generates a percent point plot.
PLOT	=	Generates a plot (including bar plots).
CLASS LOWER	=	Sets the lower class minimum for histograms, frequency plots, and pie charts.
CLASS UPPER	=	Sets the upper class maximum for histograms, frequency plots, and pie charts.
CLASS WIDTH	=	Sets the class width for histograms, frequency plots, and pie charts.
LINE	=	Sets the types for plot lines.
LINE COLOR	=	Sets the colors for plot lines.
LINE THICKNESS	=	Sets the thicknesses for plot lines.
REGION FILL	=	Sets the on/off switches for region fills.

**REFERENCE**

“The Elements of Graphing Data,” William Cleveland, Wadsworth, 1985 (p. 264).

**APPLICATIONS**

Business Graphics

**IMPLEMENTATION DATE**

The ability to set the attributes of the pie wedges was implemented 93/11.

PROGRAM

```

LET X = DATA 81 82 83 84 85
LET Y = DATA 2 5 9 15 28
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
X1LABEL SALES IN MILLIONS OF DOLLARS
.
LINE THICKNESS .3 ALL; TITLE PIE CHART WITH THICKER LINES
PIE CHART Y X
.
REGION FILL ON ALL; REGION PATTERN COLOR G10 G30 G50 G70 G90
REGION FILL COLOR G10 G30 G50 G70 G90
TITLE PIE CHART WITH SOLID FILL SLICES
PIE CHART Y X
.
TITLE PIE CHART WITH LABELS
LET N = SIZE X
LEGEND SIZE 3
LOOP FOR K = 1 1 N
    LET A = X(K)
    LEGEND ^K 19^A
END OF LOOP
LEGEND 1 COORDINATES 8 58; LEGEND 2 COORDINATES 10 71; LEGEND 3 COORDINATES 28 92
LEGEND 4 COORDINATES 68 77; LEGEND 5 COORDINATES 67 30
PIE CHART Y X
.
REGION PATTERN COLOR BLACK ALL; REGION PATTERN D1 D2 D1D2 VERT HORI
REGION PATTERN SPACING 1.0 1.0 3.0 4.0 5.0; REGION PATTERN LINE SOLID SOLID SOLID DASH DOT
TITLE PIE CHART WITH HATCH PATTERN FILLS
PIE CHART Y X
MULTIPLY OFF
    
```

